

# Graf Cayley dari suatu grup Abelian menggunakan SageMath

Oleh: Subiono, Departemen Matematika-FASAD-ITS, 2020

Suatu sumber alamia graf, dikenal sebagai **graf Cayley**, berasal dari teori grup. Teori representasi menggairahkan kita untuk menganalisis nilai-eigen dari graf Cayley setidaknya untuk grup Abelian.

**Definisi** Misalkan  $G$  adalah suatu grup berhingga. Yang dimaksud dengan **himpunan bagian simetri** dari  $G$  adalah suatu himpunan bagian  $S \subseteq G$  yang memenuhi:

1.  $1 \notin S$
2. bila  $s \in S$ , maka  $s^{-1} \in S$ .

Bila  $S$  adalah himpunan bagian simetri dari  $G$ , maka graf dari  $G$  terkait dengan  $S$  adalah graf dengan himpunan simpul  $G$  dan dengan sisi  $\{g, h\}$  menghubungkan  $g$  dengan  $h$  bila  $gh^{-1} \in S$  atau ekivalen  $hg^{-1} \in S$ .

**Catatan:** Dalam definisi  $S$  bisa himpunan kosong, dalam hal yang demikian graf Cayley tidak mempunyai sisi.

Kita bisa memferifikasi bahwa graf Cayley terhubung (setiap dua simpul terhubung oleh suatu lintasan) bila dan hanya bila  $S$  **membangun**  $G$ .

## Contoh

Misalkan  $G = \mathbb{Z}/4\mathbb{Z}$  dan  $S = \{\pm[1]_4\}$ . Untuk mendapatkan **graf Cayley** terkait dengan  $S$  dalam Sagemath sebagai berikut:

1. Buat grup Abelian  $G = \mathbb{Z}/4\mathbb{Z}$
2. Kontruksi grup permutasi  $G_p$  dari grup Abelian  $G$ .
3. Tentukan generator dari grup  $G_p$ .
4. Dari generator yang tersedia buat himpunan simetri  $S$ .
5. Terkait dengan grup  $G_p$ , buat matriks ketetanggaan  $A$  dengan generator  $S$ .
6. Dari matriks ketetanggaan  $A$  buat graf Cayleynya.

```
In [2]: G = AdditiveAbelianGroup([4])
Gp = G.permutation_group()
g0= Gp.gens()
g1=g0[0]
S=[g1,g1^-1]
S1=set(S)
print("Himpunan simetri: S =",S1)
```


Himpunan simetri: S = {(1,4,3,2), (1,2,3,4)}

```
In [3]: A = Gp.cayley_graph(generators=S, simple=True).adjacency_matrix()
pretty_print(html("Matriks ketetanggaan $A=%s$" % latex(A)))
```

$$\text{Matriks ketetanggaan } A = \begin{pmatrix} 0 & 1 & 0 & 1 \\ 1 & 0 & 1 & 0 \\ 0 & 1 & 0 & 1 \\ 1 & 0 & 1 & 0 \end{pmatrix}$$

```
In [4]: Gamma = Graph(A)
pretty_print(html("Graf $\Gamma$ dari matriks ketetanggaan $A$ :"))
Gamma.show(layout="circular", dpi = 80)
```

Graf  $\Gamma$  dari matriks ketetanggaan  $A$  :



```
In [ ]:
```

Perintah SageMath untuk menghasilkan jenis **graf Cayley** sedikit rumit. Kita pasti ingin memastikan untuk memilih opsi "sederhana" dan untuk memilih himpunan pembangun simetris, sebagaimana diberikan oleh dua contoh graf Cayley dari grup Abelian  $\mathbb{Z}/6\mathbb{Z}$  untuk himpunan simetri  $S_1 = \{\pm 1, \pm 2\}$  dan  $S_2 = \{\pm 1, \pm 3\}$  sebagai berikut berikut:

```
In [4]: G = AdditiveAbelianGroup([6])
Gp = G.permutation_group()
g1, g2 = Gp.gens()
S1 = [g1*g2, (g1*g2)^-1, (g1*g2)^2, (g1*g2)^-2]
print("Himpunan Simetri S1 =", set(S1))
```

Himpunan Simetri  $S_1 = \{(3,5,4), (1,2)(3,4,5), (3,4,5), (1,2)(3,5,4)\}$


```
In [5]: A1 = Gp.cayley_graph(generators=S1, simple=True).adjacency_matrix()
```

```
In [6]: pretty_print(html(" Matriks ketetanggaan $A_1 = %s$"%latex(A1)))
```

$$\text{Matriks ketetanggaan } A_1 = \begin{pmatrix} 0 & 1 & 1 & 0 & 1 & 1 \\ 1 & 0 & 1 & 1 & 0 & 1 \\ 1 & 1 & 0 & 1 & 1 & 0 \\ 0 & 1 & 1 & 0 & 1 & 1 \\ 1 & 0 & 1 & 1 & 0 & 1 \\ 1 & 1 & 0 & 1 & 1 & 0 \end{pmatrix}$$

```
In [7]: Gamma1 = Graph(A1)
pretty_print(html("Graf $\Gamma_1$ dari matriks Adjacency $A_1$ :"))
Gamma1.show(layout="circular", dpi = 60)
```

Graf  $\Gamma_1$  dari matriks Adjacency  $A_1$  :



```
In [8]: G = AdditiveAbelianGroup([6])
Gp = G.permutation_group()
g1,g2 = Gp.gens()
S2=[g1*g2,(g1*g2)^-1,(g1*g2)^3,(g1*g2)^-3]
print("Himpunan Simetri S2 = ",set(S2))
```

Himpunan Simetri S2 =  $\{(1,2)(3,4,5), (1,2), (1,2)(3,5,4)\}$


```
In [9]: A2 = Gp.cayley_graph(generators=S2,simple=True).adjacency_matrix()
```

```
In [10]: pretty_print(html(" Matriks Ketetanggaan $A_2 = %s$"%latex(A2)))
```

$$\text{Matriks Ketetanggaan } A_2 = \begin{pmatrix} 0 & 0 & 0 & 1 & 1 & 1 \\ 0 & 0 & 0 & 1 & 1 & 1 \\ 0 & 0 & 0 & 1 & 1 & 1 \\ 1 & 1 & 1 & 0 & 0 & 0 \\ 1 & 1 & 1 & 0 & 0 & 0 \\ 1 & 1 & 1 & 0 & 0 & 0 \end{pmatrix}$$


```
In [11]: Gamma2 = Graph(A2)
pretty_print(html("Graf  $\Gamma_2$  dari matriks Ketetanggaan  $A_2$  :"))
Gamma2.show(layout="circular", dpi = 60)
```

Graf  $\Gamma_2$  dari matriks Ketetanggaan  $A_2$  :



Selanjutnya dibahas graf Cayley  $\Gamma$  dari  $\mathbb{Z}/4\mathbb{Z} \times \mathbb{Z}/4\mathbb{Z}$ , dengan himpunan generator  $S = \{\pm(0, 1), \pm(1, 0), \pm(1, 1)\}$ . Kita implementasikan hal ini dalam SageMath sebagai berikut:

```
In [12]: G = AdditiveAbelianGroup([4,4])
GP = G.permutation_group()
g0 = GP.gens()[0]
g1 = GP.gens()[1]
S = [g0,g1, g0^(-1), g1^(-1),(g0*g1)^(-1),g0*g1]
Gamma = GP.cayley_graph(side='left', generators = S)
A = Gamma.adjacency_matrix()
Gamma1 = Graph(A, format = "adjacency_matrix")
Gam = Gamma1.show(layout="circular", dpi = 70)
pretty_print(html("Graf Cayley dari \
 $\mathbb{Z}/4\mathbb{Z} \times \mathbb{Z}/4\mathbb{Z}$ \
yang dibangun oleh  $S$ "))
Gam
```



Graf Cayley dari  $\mathbb{Z}/4\mathbb{Z} \times \mathbb{Z}/4\mathbb{Z}$  yang dibangun oleh  $S$

In [ ]:

Berikut ini diberikan graf Cayley dari grup  $\mathbb{Z}/24\mathbb{Z}$  dengan himpunan simetri  $S = \{[8]_{24}, [16]_{24}, [3]_{24}, [21]_{24}\}$ .

```
In [13]: G = AdditiveAbelianGroup([24])
Gp = G.permutation_group()
g1,g2 = Gp.gens()
S2 = [g1,g1^-1,g2,g2^-1]; S2
```

```
Out[13]: [(4,5,6,7,8,9,10,11), (4,11,10,9,8,7,6,5), (1,2,3), (1,3,2)]
```

```
In [14]: A2 = Gp.cayley_graph(generators=S2,simple=True).adjacency_matrix()
show(A2)
```


```

(
0 1 0 0 0 0 0 1 1 0 0 0 0 0 0 1 0 0 0 0 0 0
1 0 1 0 0 0 0 0 0 1 0 0 0 0 0 0 1 0 0 0 0 0
0 1 0 1 0 0 0 0 0 0 1 0 0 0 0 0 0 1 0 0 0 0
0 0 1 0 1 0 0 0 0 0 0 1 0 0 0 0 0 0 1 0 0 0
0 0 0 1 0 1 0 0 0 0 0 0 1 0 0 0 0 0 0 1 0 0
0 0 0 0 1 0 1 0 0 0 0 0 0 1 0 0 0 0 0 0 1 0
0 0 0 0 0 1 0 1 0 0 0 0 0 0 1 0 0 0 0 0 0 1
1 0 0 0 0 0 1 0 0 0 0 0 0 0 0 1 0 0 0 0 0 0
1 0 0 0 0 0 0 0 0 1 0 0 0 0 0 1 1 0 0 0 0 0
0 1 0 0 0 0 0 0 1 0 1 0 0 0 0 0 1 0 0 0 0 0
0 0 1 0 0 0 0 0 0 1 0 1 0 0 0 0 0 1 0 0 0 0
0 0 0 1 0 0 0 0 0 0 1 0 1 0 0 0 0 0 1 0 0 0
0 0 0 0 1 0 0 0 0 0 0 1 0 1 0 0 0 0 0 1 0 0
0 0 0 0 0 1 0 0 0 0 0 0 1 0 1 0 0 0 0 0 1 0
0 0 0 0 0 0 1 0 0 0 0 0 0 1 0 1 0 0 0 0 0 1
0 0 0 0 0 0 0 1 1 0 0 0 0 0 1 0 0 0 0 0 0 0
1 0 0 0 0 0 0 0 1 0 0 0 0 0 0 0 1 0 0 0 0 0
0 1 0 0 0 0 0 0 0 1 0 0 0 0 0 0 1 0 1 0 0 0
0 0 1 0 0 0 0 0 0 0 1 0 0 0 0 0 0 1 0 1 0 0
0 0 0 1 0 0 0 0 0 0 0 1 0 0 0 0 0 1 0 1 0 0
0 0 0 0 1 0 0 0 0 0 0 0 1 0 0 0 0 0 1 0 1 0
0 0 0 0 0 1 0 0 0 0 0 0 0 1 0 0 0 0 0 1 0 1
0 0 0 0 0 0 1 0 0 0 0 0 0 0 1 0 0 0 0 0 1 0
0 0 0 0 0 0 0 1 0 0 0 0 0 0 0 1 1 0 0 0 0 1
)

```

```
In [16]: Gamma2 = Graph(A2)
pretty_print(html("Graf  $\Gamma$  dari matriks Ketetanggaan  $A_2$  :"))
Gamma2.show(layout="circular", dpi = 60)
```

Graf  $\Gamma$  dari matriks Ketetanggaan  $A_2$  :



### Sebagai penutup dari tulisan ini, diberikan beberapa komentar:

1. Pembahasan graf Cayley yang dibahas untuk grup komutatif/Abelian  $G$ .
2. Dari grup komutatif yang diberikan ditentukan grup permutasinya, yaitu  $G_p$ .
3. Dari grup permutasi yang telah didapat ditentukan generator-generatornya.
4. Selanjutnya dari generator yang tersedia dibentuk himpunan simetri  $S$ . Tentunya dalam hal ini  $S$  tidak tunggal.
5. Selanjutnya dibuat matriks ketetanggaan  $A$  dari grup permutasi berdasarkan generator  $S$ .
6. Dibuat graf Cayley dari grup komutatif  $G$  berdasarkan matriks ketetanggaan  $A$ .
7. Apakah untuk grup non-Abelian (tidak komutatif) bisa dibuat graf Cayleynya? Jawabannya bisa hal ini terkait dengan pelabelan graf pada sisi-sisinya.

## TERIMA KASIH ATAS PERHATIANNYA! SEMOGA BERMANFAAT.

```
In [ ]: # Produk Langsung dan Grup Abelian

from IPython.display import Audio, Image, YouTubeVideo
id = 'HSX69ByJppo'
YouTubeVideo(id=id, width=600, height=500)
```

```
In [5]: A=[1,2,3,4,5,6,7,8]
A
```

```
Out[5]: [1, 2, 3, 4, 5, 6, 7, 8]
```

```
In [9]: G = AdditiveAbelianGroup([24])
        Gp = G.permutation_group()
        g1,g2 = Gp.gens()
        S2 = [g1,g1^-1,g2,g2^-1]; S2
```

```
Out[9]: [(4,5,6,7,8,9,10,11), (4,11,10,9,8,7,6,5), (1,2,3), (1,3,2)]
```

```
In [11]: g1.order()
```

```
Out[11]: 8
```

```
In [12]: g2.order()
```

```
Out[12]: 3
```

```
In [13]: g2^3
```

```
Out[13]: ()
```

```
In [ ]:
```